

**Przedmiot:** Informatyka w inżynierii produkcji

**Forma:** Laboratorium

**Temat:** Zadanie 3. Definicje zmiennych i stałych. Manipulacje zmiennymi typu String.

Celem ćwiczenia jest nabycie umiejętności definiowania zmiennych i stałych na poziomie procedur oraz zapoznanie się z użytecznymi funkcjami pozwalającymi na manipulacje zmiennymi typu String w praktyce (separacja, łączenie tekstu, znajdowanie znaków w tekście, zmiana wielkości liter).

### Definicja zmiennych

**Dim** NazwaZmiennej [**As** TypDanych]

### Definicja stałych

**Const** NAZWA\_STALEJ [**As** TypDanych] = Wartość

Wartości *stałych* są ustalane raz z momencie deklaracji i nie można ich zmienić. Wartości *zmiennych* można zmieniać dowolną ilość razy.

### Podstawowe funkcje do manipulacji łańcuchem tekstu:

**Łączenie napisów:** & (różnych typów), +

**Str(x)** – zamiana liczby typu Long na String

**UCase(znaki)** – zamiana ciąg znaków na wielkie litery

**LCase(znaki)** – zamiana ciąg znaków na małe litery

**LTrim(znaki)** – zwraca ciąg znaków bez spacji początkowych

**RTrim(znaki)** – zwraca ciąg znaków bez spacji końcowych

**Trim(znaki)** – zwraca ciąg znaków bez spacji początkowych i końcowych

**Left(znaki,ile\_znakow)** – zwraca ciąg znaków zawierający początkowe *ile\_znakow* zmiennej *znaki*

**Right(znaki)** – zwraca ciąg znaków zawierający końcowe *ile\_znakow* zmiennej *znaki*

**Mid(znaki, start [,ile\_znakow])** – zwraca ciąg znaków zawierający *ile\_znakow* zmiennej *znaki*, począwszy od pozycji *start*

**Len(znaki)** – zwraca liczbę znaków zmiennej *znaki*

**InStr(s,s1)** - znajduje pozycję w łańcuchu *s*, na której znajduje się pierwsze wystąpienie podłańcucha *s1*; jeśli podłańcuch *s1* w ogóle nie występuje w *s* to *InStr* zwraca wartość 0.

**InStr(k,s,s1)** - znajduje pozycję w łańcuchu *s*, na której znajduje się podłańcuch *s1*, przy czym przeszukiwanie rozpoczyna się od pozycji *k*.

## ZADANIE 1.

W arkuszu „Zadanie1” (jak poniżej) dodaj przycisk „Rozdziel dane”, który spowoduje uzupełnienie kolumn *Imię* i *Nazwisko*, danymi pochodzącymi z kolumny *Pracownik*.

	A	B	C	D	E	F	G
1	<b>Pracownik</b>	<b>Imię/Imiona</b>	<b>Nazwisko</b>				
2	Adam, Kowalski						
3	Janusz Andrzej, Maciąg						
4	Karina, Sowa						
5	Jan, Grolla						
6	Jakub, Lotto						
7	Karolina, Olaf-Kola						
8	Monika Joanna, Słoń						
9	Wiesława, Lutta						
10							

Rozdziel dane

## Rozwiązanie.

Do przycisku „Rozdziel dane” przypisz procedurę o nazwie „RozdzielDane\_Click”. Przykład procedury zamieszczono poniżej.

```
Sub RozdzielDane_Click()  
    'definicja zmiennych  
    Dim imie As String  
    Dim nazwisko As String  
    Dim pracownik As String  
    Dim pozP As Integer 'pozycja przecinka  
    Dim wiersz As Integer  
  
    For wiersz = 2 To 9  
        pracownik = Cells(wiersz, 1).Value  
        'odszukanie pozycji na której jest ","  
        pozP = InStr(pracownik, ",")  
        'wyciągnięcie imion i nazwisk  
        imie = Left(pracownik, pozP - 1)  
        nazwisko = Mid(pracownik, pozP + 2) 'od pozycji pozP+2 do końca  
        'przepisanie danych do komórek  
        Cells(wiersz, 2) = imie  
        Cells(wiersz, 3) = nazwisko  
    Next wiersz  
End Sub
```

## ZADANIE 2. (do samodzielnego rozwiązania)

W arkuszu Zadanie2! (jak poniżej) dodaj przycisk „Podziel adres”, który spowoduje uzupełnienie kolumn *Miasto*, *Kod pocztowy*, *Adres* danymi pochodzącymi kolumny *Adres dostawcy*. Kolumna adres ma zawierać ulicę (bez przedrostka ul.) oraz nr domu/mieszkania.

	A	B	C	D	
1	<b>Adres dostawcy</b>	<b>Miasto</b>	<b>Kod pocztowy</b>	<b>Adres</b>	
2	45-090 Kędzierzyn-Koźle; ul. Wiśniowa 78				
3	45-190 Będzin; ul. Opolska 4/7				
4	77-090 Wielkopolska; ul. Oleska 90				
5	67-068 Kędzierzyn-Koźle; ul. Ogrodowa 1/12				
6	44-099 Opole; ul. Wiśniowa 5				
7	67-888 Wrocław; ul. Brzeska 3				
8	45-090 Poznań; ul. Wołyńska 84				
9					
10					

### ZADANIE 3.

W arkuszu „Zadanie3” (jak poniżej) dodaj przycisk „**Poprawna wielkość czcionki**”, który spowoduje zmianę wielkości czcionek na poprawną w kolumnach *Imię*, *Nazwisko*.

	A	B	C	D
1	<b>Lp</b>	<b>Imię</b>	<b>Nazwisko</b>	
2	1	ADAM	kowalski	
3	2	ANDrzej	Maciąg	
4	3	karina	Sowa	
5	4	Jan	grolla	
6	5	Jakub	Lotto	
7	6	karolina	OlafF	
8	7	Monika	SŁOŃ	
9	8	Wiesława	Lutta	
10				

### Rozwiązanie.

Do przycisku „Poprawna wielkość czcionki” przypisz procedurę o przykładowej nazwie „WielkoscCzcionki\_Click”. Przykład procedury zamieszczono poniżej.

```
Sub WielkoscCzcionki_Click()  
    'definicja zmiennych  
    Dim imie As String  
    Dim nazwisko As String  
    Dim wiersz As Integer  
  
    For wiersz = 2 To 10000  
        'wyjdź z pętli jeżeli jest pusta komórka w pierwszej kolumnie  
        If Cells(wiersz, 1) = "" Then  
            Exit For  
        End If  
  
        'zmień wielkość czcionek  
        imie = UCase(Left(Trim(Cells(wiersz, 2)), 1)) & LCase(Mid(Cells(wiersz, 2), 2))  
        nazwisko = UCase(Left(Trim(Cells(wiersz, 3)), 1)) & LCase(Mid(Cells(wiersz, 3), 2))  
        'przepisanie danych do komórek  
        Cells(wiersz, 2) = imie  
        Cells(wiersz, 3) = nazwisko  
  
    Next wiersz  
End Sub
```

Zadanie można również rozwiązać używając bezpośrednio funkcji:

*Imie = WorksheetFunction.Proper(Cells(wiersz, 2))*

### ZADANIE 4. (do samodzielnego rozwiązania)

W arkuszu „Zadanie4” (jak po prawej) dodaj przycisk „**Poprawna wielkość czcionki**”, który spowoduje zmianę wielkości czcionek na poprawną w kolumnach *Imię/Imiona*, *Nazwisko* (zwróć uwagę, że bywają imiona i nazwiska podwójne).

	A	B	C	D
1	<b>Lp</b>	<b>Imię/Imiona</b>	<b>Nazwisko</b>	
2	1	ADAM	kowalski	
3	2	ANDrzej	Maciąg	
4	3	karina UrszULA	Sowa-NOWak	
5	4	Jan	grolla	
6	5	Jakub	Lotto	
7	6	karolina ANNA	Olaff-JOrch	
8	7	Monika	SŁOŃ	
9	8	Wiesława	Lutta	

### ZADANIE 5.

W arkuszu „Zadanie5” dodaj przycisk „Zlicz znaki”, który spowoduje wyświetlenie komunikatu (*MsgBox*) z liczbą znaków, jaka jest zawarta w zaznaczonym zakresie komórek (*Range*). Przed uruchomieniem przycisku należy zaznaczyć zakres komórek, w których chcemy zliczyć znaki.

### Rozwiązanie.

Do przycisku „Zlicz znaki” przypisz procedurę o przykładowej nazwie „ZliczZnaki\_Click”. Przykład procedury zamieszczono poniżej.

```
Sub ZliczZnaki_Click()  
    'definicja zmiennych i stałej  
    Dim iloscZnakow As Long  
    Dim zakres As Range  
    Dim komorka As Range  
    Const komunikat = "Ilość znaków w zaznaczonym zakresie to: "  
  
    'przypisanie do zmiennej zakres zaznaczonego zakresu komórek  
    Set zakres = Selection  
    iloscZnakow = 0  
  
    'zliczanie liczby znaków w kolejnych komórkach  
    For Each komorka In zakres  
        iloscZnakow = iloscZnakow + Len(komorka.Value)  
    Next komorka  
  
    MsgBox (komunikat & iloscZnakow)  
End Sub
```

### ZADANIE 6. (dodatkowe)

W arkuszu „Zadanie5” dodaj przycisk „Zlicz wyrazy”, który spowoduje wyświetlenie komunikatu (*MsgBox*) z liczbą wyrazów, jaka jest zawarta w zaznaczonym zakresie komórek (*Range*). Przed uruchomieniem przycisku należy zaznaczyć zakres komórek, w których chcemy zliczyć wyrazy.